
Motivation to 
develop a cross-

report verification  
system



 We were working on a large phase III clinical trial for 
HIV prevention with requirement for bi-annual Data 
Safety and Monitoring Reports (DSMB).

 Faced with verifying hundreds of pages of tables, 
listings and figures (TLFs) under tight timelines.

 Manual review of TLFs is very time-consuming and 
error-prone. 

 Reports were growing in size over time and frequency 
of reporting was enough to warrant investing in the 
development of an automated system.

Motivation



General framework 
for the cross-

report verification 
system



 We developed procedures utilizing SAS® to automate a 
verification checks, improving the accuracy and 
efficiency of the TLFS for DSMB reports

 Results data sets used to generate TLFs

 Excel driver file created and contained verification checks 
to be performed

 Verification checks developed and performed on the 
results datasets

 HTML-formatted file developed to show what errors 
occurred and where they occurred

General Framework



General Framework



Results Data Set



 Standard data sets are created for each TLF 
through the use of SAS® macros

Data sets are concatenated to form one large 
results data set 

 Standardized format and contains enough 
information to make every element of the 
TLFs uniquely identifiable

Basis for the verification checks

Features of the Results Data Set



 prgmid (program ID): Directory location and file name of SAS® 

program 
 dataset_created_from: Unique identifier representing name of 

the standardized data set based on the TLF
 number_of_titles, number_of_footnotes: Number of titles & 

footnotes from a TLF
 title1-title10, footnote1-footnote10: Values of title1-title10 and 

footnote1-footnote10 from a TLF
 rowvar (row variable description): Row heading in a table
 colvar (column variable description): Column heading in a table
 repvar (report variable): Value of a cell in a table (or data 

element in a figure)

Variables in the Results Data Set



Example of the Results Data Set

 prgmid = 
/open/t_dem_site_blind.sas

 dataset_created_from = 
demout_site

 number_of_titles = 4
 title1-title4 = 
 title1 = A Phase III Clinical Trial
 title2 = DSMB Open Report –

January 1, 9999
 title3 = Visit Cutoff Date: April 

1, 2015 
 title4 = Table 2. Demographics 

of Enrolled Participants by Site



Example of the Results Data Set

First row of the Results Data Set:

 rowvar = Participants Enrolled

 colvar = Site 1

 repvar = 157

 Corresponds to the 
Participants Enrolled row 
and Site 1 column in the 
table

rowvar



Structure of the Results Data Set



Structure of the Results Data Set

 First box corresponds to the format the data takes in the 
Results Data Set



Structure of the Results Data Set

 Second box corresponds to the output as it appears in the 
table used for the report



Details of the SAS®

Code



Step 1. Create the Excel 
Verification Driver File



Manually created to define the verification 
checks to be made for each report

Features of the Excel Verification 
Driver File (verification_checks.xlsx)

Verification check ID Description Status Notes

VC_0001

Total Enrolled from Accrual Summary by Site table 

should match Participants Enrolled when it occurs in 

other tables.

VC_0002 Title 1 should match for all pages.

VC_0003 Title 2 should match for all pages in open report.

VC_0004 Title 2 should match for all pages in closed report.

VC_0005 Title 3 should match for all pages.

VC_0006 Check the SAS logs

PROC IMPORT 

out=checks
datafile="Verification_checks.xlsx" 

dbms=excelcs replace;    

run;

Create a SAS® dataset from the Excel 
Verification Driver File



DATA temp;

attrib Status length=$20 format=$20. 

Notes length=$256 format=$256; 

set checks(where=(^missing(Verification_check_id))); 

Status = 
strip(Status)||'&'||strip(Verification_check_id)||'_status';

Notes = 
strip(Notes)||'&'||strip(Verification_check_id)||'_notes';

run;

Create “temp” dataset that adds 
macro variables to Status & Notes 



Example of “temp” dataset with 
added macro variables to Status &

Notes  



ODS HTML file= "Verification_template.htm";

PROC REPORT data=temp headline headskip nowd ls=256 
center style(report)=table[frame=below rules=rows];

column Obs Verification_check_ID Description Status Notes;

define Obs / 'Obs' center;
define Verification_check_ID / 

'Verification Check ID' center order  style(column)=[cellwidth=2in just=center];
define Description / 

'Description' flow width=50 style(column)=[cellwidth=5in just=left] ;
define Status /  'Status' flow style(column)=[cellwidth=1in just=center];
define Notes /   'Notes' style(column)=[cellwidth=4in just=left]  flow;
compute obs;
cntl+1;
obs = cntl;

endcomp;
run;
ODS HTML close;

Create HTML Template File using 
Proc Report and ODS HTML



Example of the HTML Template File



Step 2. Run the Verification Checks 
against the Results Data Set



 The Verification Checks are detailed in the Excel Verification 
Driver File. 

 Each program has a specific verification aim, such as confirming 
that the enrollment totals match across all TLFs. 
 Accomplished by comparing all occurrences of the relevant 

data element within the Results Data Set and determining 
whether the values of those cells satisfy the conditions of the 
check. 

 During this process the values for the macro variables 
&VC_<IDnumber>_status and &VC_<IDnumber>_notes are 
created and stored. 

Create separate SAS® programs for 
each of the Verification Checks 



 Total enrollment generally appears in tables in a row with the 
heading of “Participants Enrolled”, and occurs in a column 
corresponding to All Sites.

 Using the Results Data Set, called work.save_all, subset on the 
relevant observations for this check where the rowvar value equals 
“Participants Enrolled” and the colvar value equals ’All Sites’ = 999.

DATA VC_0001;

set work.save_all;
if upcase(strip(ROWVAR)) =: 'PARTICIPANTS ENROLLED‘ 
and colvar = 999;

run ;

Example of a Verification Check of 
the enrollment numbers 



 Create a new variable called num_repvar as a 
numeric version of repvar, a character variable in 
the Results Data Set

data all ;
length num_repvar 8;
set VC_0001  end=eof ;

num_repvar = input(strip(repvar),10.) ;
keep prgmid varchr colvar num_repvar repvar ;
run ;

Example of a Verification Check of 
the enrollment number



 Create a PDF file that summarizes the enrollment numbers 
found in the Results Data Set

ODS pdf file = "VC_0001.pdf" bookmarklist=hide notoc;
title "VC_0001: Total Enrolled from Accrual Summary by Site table should match 
Participants Enrolled when it occurs in other tables";

PROC REPORT data=all nowd ls=256;
column prgmid varchr colvar num_repvar;

define prgmid / 'Source File' display style(column)=[cellwidth=6in just=left];
define varchr / 'Description' display style(column)=[cellwidth=1.5in just=right];
define colvar / 'Column' display style(column)=[cellwidth=1in  just=right];
define num_repvar / 'Result' display style(column)=[cellwidth=1in  just=right];

run;
ODS pdf close; 

Example of a Verification Check of 
the enrollment numbers 



Example of the PDF file created from 
the Verification Check



 The check compares the minimum and maximum values 
of the num_repvar variables using Proc SQL

 If the minimum and maximum are equal then all values 
must be identical so it will pass, otherwise it fails  

PROC SQL noprint;
select min(num_repvar)- max(num_repvar) into :diff  

from VC_0001;
quit;

Example of a Verification Check of 
the enrollment numbers 



 If the minimum and maximum are equal then an %IF statement gives the 
&VC_0001_status macro variable the value of “Passed”, otherwise it gives 
the value of “Failed”. 

 Assign an HTML link containing a path to the PDF summary file

%MACRO check;

%global VC_0001_status VC_0001_notes;

%let VC_0001_notes = &notes;

%if &diff = 0 %then %let VC_0001_status=<A HREF="VC_0001.pdf">Passed</A>;

%else %let VC_0001_status=<A HREF="VC_0001.pdf">Failed</A>;

%MEND check;

%check;

Example of a Verification Check of 
the enrollment numbers 



Keep in mind the results as seen in 
the PDF file



Step 3. Create an HTML Report File 
with the result of the verification 

checks



 Once all of the verification checks specified in the Excel 
Verification Driver File have run, we update our HTML Template 
File and make the Validation Dashboard.

 We update it with the macro variables 
&VC_<IDnumber>_status and &VC_<IDnumber>_notes stored 
for each check as a global macro variables

Update the HTML Template File and 
make the Validation Dashboard



%MACRO read_htm (webin,webout);
filename web_in "&webin";
filename web_out "&webout";
DATA _null_;

length line $ 256;
infile web_in length=lvg;
file web_out;
input @1 line $varying200. lvg;
line=tranwrd(line,'&amp;','&'); 
line = trim(resolve(line));
put line;

run;
filename web_in; 
filename web_out; 

%MEND read_htm;

%read_htm (webin= Validation_template.htm, webout=Validation_dashboard.htm ); 

Update the HTML Template File and 
make the Validation Dashboard



%MACRO read_htm (webin,webout);
filename web_in "&webin";
filename web_out "&webout";
DATA _null_;

length line $ 256;
infile web_in length=lvg;
file web_out;
input @1 line $varying200. lvg;
line=tranwrd(line,'&amp;','&'); 
line = trim(resolve(line));
put line;

run;
filename web_in; 
filename web_out; 

%MEND read_htm;

%read_htm (webin= Validation_template.htm, webout=Validation_dashboard.htm ); 

Update the HTML Template File and 
make the Validation Dashboard

Required since 
HTML code shows 
&amp; instead of & 



%MACRO read_htm(webin,webout);
filename web_in "&webin";
filename web_out "&webout";
DATA _null_;

length line $ 256;
infile web_in length=lvg;
file web_out;
input @1 line $varying200. lvg;
line=tranwrd(line,'&amp;','&'); 
line = trim(resolve(line));
put line;

run;
filename web_in; 
filename web_out; 

%MEND read_htm;

%read_htm (webin= Validation_template.htm, webout=Validation_dashboard.htm ); 

Update the HTML Template File and 
make the Validation Dashboard

Resolves the macro 
variables defined as:
VC_0001_status=<A 
HREF="VC_0001.pdf">Passe
d</A> or VC_0001_status=<A 
HREF="VC_0001.pdf">Failed
</A>;



Example of the Validation Dashboard 



 Verification checks are made on the SAS® data sets used to create the TLFs, 
rather than on the text contained in the PDF output of the TLFs 
themselves. 

 If one of the TLFs in the report is created by a SAS® program that is not 
included in the checks for that report, or if the data set that creates that 
TLF is not part of the Results Data Set, then it is possible for discrepancies 
from that TLF to go unnoticed. 

 Issues can also arise that we did not anticipate when writing the logic of 
the verification checks, and which subsequently are not caught by the 
automated process. Therefore, even when the checks identify failures, 
manual effort is still required to identify the source of those failures and to 
make the necessary corrections.

Limitations 



 Proved to be extremely helpful in completing the review process 
necessary to produce large and complex DSMB reports with high 
accuracy and integrity. 

 By focusing on automating verification checks that were simple as 
well as those that would have been time consuming to do manually, 
we were able to focus our attention on the more complicated 
aspects of preparing the reports. 

 Finally, we have incorporated the SAS® automated cross-report 
verification system into our daily crons and it has also proven to be 
useful in identify programming bugs and other problems early on in 
our daily reporting process. 

Conclusions



Thank you!


